

H₂O-CO₂-NaCl系流体の密度および逸散能 (fugacity) を求めるための BASIC プログラム

澁江 靖 弘*

(平成6年9月20日受理)

1. はじめに

地殻中に存在する流体の主成分が、多くの場合、H₂O、CO₂、NaClであることが流体包有物の分析 (例えば、ROEDDER, 1984) などから明らかになっている。すると、この三成分系流体 (H₂O-CO₂-NaCl) に関する逸散能 (fugacity) などの熱力学的性質が岩石と流体との反応を考える上で重要になってくる。また、流体包有物の組成と均質化温度から流体の温度を求めようとすると、流体の密度を求めておく必要がある (例えば、澁江, 1993)。この三成分系流体の逸散能や密度に関して、BOWERS and HELGESON (1983) は半経験的な理論式を求めた。また、FORTRAN 言語で書かれたコンピュータプログラムも公表している (BOWERS and HELGESON, 1985)。筆者は、このプログラムを対話型言語である BASIC に書き直し、ユーザーがより利用しやすいものにした (澁江, 1990)。本報告では、筆者が作成したプログラムのリストを示し、その使用法を記述する。なお、BOWERS and HELGESON (1983) が用いた式については別に解説する (澁江, 1995) のでここでは詳しくは触れないことにする。

2. 状態方程式

BOWERS and HELGESON (1983) は、まず、NaCl が高温条件下では電離せずに会合分子 (電気的には中性) になることを利用して、この三成分系流体を気体混合物として取り扱った。そして、REDLICH and KWONG (1949) が提唱した状態方程式を用いた。

$$P = RT / (V - b) - a / T^{0.5} V(V + b) \quad (1)$$

式(1)中のaとbは経験的なパラメーターであり、Pは圧力、Tは絶対温度、Rは気体定数、Vはモル体積を表している。REDLICH and KWONG (1949)の式ではaとbは定数である。H₂O、CO₂あるいはH₂O-CO₂系についてのRedlich-Kwong型の状態方程式が既にいくつか求められており (FERRY and BAUMGARTNER, 1987)、この中でCO₂とH₂O-CO₂系に関するDE SANTIS et al. (1974)とHOLLOWAY (1977)の式がそのまま用いられている。そして、GEHRIG (1980)の三成分系に関する温度-圧力-体積の実験結果を基にしてH₂O-CO₂-NaCl系に関する経験的パラメーター (式(1)中のaとb)が温度と組成の関数として求められている。このaとbは次のように表された。

分系に関する温度-圧力-体積の実験結果を基にしてH₂O-CO₂-NaCl系に関する経験的パラメーター (式(1)中のaとb)が温度と組成の関数として求められている。このaとbは次のように表された。

$$a = X_{CO_2}^2 a_{CO_2} + (1 - X_{CO_2})^2 a_{H_2O} + 2X_{CO_2}(1 - X_{CO_2}) \{ (a_{CO_2} a_{H_2O})^{0.5} + 0.5R^2 T^{2.5} K \} \quad (2)$$

$$b = X_{CO_2} b_{CO_2} + (1 - X_{CO_2}) b_{H_2O} \quad (3)$$

式(2)と(3)中のX_{CO₂}はCO₂のモル分率であり、他のパラメーターは以下のように表せる。

$$a_{CO_2} = (73.03 - 0.0714t + 2.157 \cdot 10^{-5}t^2) \cdot 10^6 \quad (4)$$

$$a_{H_2O} = 111.3057 + 50.70033 \exp(-9.82646 \cdot 10^{-3}T) + w \{-8.05658 \exp(-9.82646 \cdot 10^{-3}T)\} \quad (5)$$

$$a_{CO_2} = 4.6 \cdot 10^7 \quad (6)$$

$$a_{H_2O} = 10^6 \exp(\Theta + w \Xi + w^2 T + w^3 \Phi) \quad (7)$$

$$\ln K = -11.071 + 5953/T - 2.746 \cdot 10^6/T^2 + 4.646 \cdot 10^8/T^3 \quad (8)$$

$$b_{CO_2} = 29.7 \quad (9)$$

$$b_{H_2O} = 14.6 - 0.04420283w \quad (10)$$

式(4)中のtは摂氏温度を示し、式(5)と式(7)と式(10)中のwはH₂Oに溶解しているNaClの重量%濃度を表している。NaClがCO₂に殆ど溶解しないことから、このwの値は系全体中のNaClのモル分率から計算することができる。具体的には次式でwを計算する。H₂OとNaClの分子量がそれぞれ18.016と58.448であることを用いると、 $w = 100 \cdot X_{NaCl} \cdot 58.448 / (X_{NaCl} \cdot 58.448 + X_{H_2O} \cdot 18.016)$ である。また、 Θ 、 Ξ 、 T 、 Φ はそれぞれ次式のように与えられている。

$$\Theta = 4.881243 + 1.823047 \cdot 10^{-3}T - 1.712269 \cdot 10^{-5}T^2 + 6.479419 \cdot 10^{-9}T^3 \quad (11)$$

$$\Xi = 2.636494 \cdot 10^{-2} - 5.36994 \cdot 10^{-4}T + 2.687074 \cdot 10^{-6}T^2 - 4.321741 \cdot 10^{-9}T^3 \quad (12)$$

$$T = 6.802827 \cdot 10^{-3} - 9.48023 \cdot 10^{-5}T + 3.770339 \cdot 10^{-7}T^2 - 5.075318 \cdot 10^{-10}T^3 \quad (13)$$

$$\Phi = 5.235827 \cdot 10^{-5} - 3.505272 \cdot 10^{-8}T \quad (14)$$

さて、成分iの逸散能 (f_i)は、系全体の圧力 (P)、成分iのモル分率 (X_i)、逸散能係数 (χ_i)を用いて、f_i = P X_i χ_iと表せる。そして、χ_iは式(1)と式(2)中のaとbの値より次式で求めることができる。

$$\ln \chi_i = \ln \{ V / (V - b) \} + \{ (\partial n_b / \partial n_i) \} / (V - b)$$

*兵庫教育大学第3部(自然系教育講座)

$$- [\ln\{(V+b)/V\}/RT^{1.5}n_i b] (\partial n_i^2 a / \partial n_i) + [n_i^2 a \{ \partial n_i^2 b / \partial n_i \} / RT^{1.5} n_i b^2] [\ln(V+b)/V - b/(V+b)] - \ln(Z) \quad (15)$$

上式中の n_i は系全体のモル数であり、式中に現れる偏微分式はすべて温度、圧力、 i 以外の成分のモル数が一定の条件下におけるものである。また Z は $Z = PV/RT$ で定義される(圧縮係数と呼ばれる)値である。偏微分の項は、式(2)と式(3)より変形して求めることができる。長くなるので割愛するが、原報の式(16)から(22)に示されている。

さて、式(15)中の V と圧縮係数(Z)との間には次の関係式が成立する。

$$V = ZRT/P \quad (16)$$

すると、 Z を求めることができれば、式(16)より体積を求めることができ、この体積の計算値と式(2)と式(3)から計算できる a 、 b を式(15)に代入すれば逃散能係数が求められる。また、流体の組成から混合流体1モルあたりの質量が計算できるので、1モルあたりの体積を用いて密度を求めることもできる。

さて Z の求め方に戻る。 Z は次の三次方程式の解に相当する。

$$Z^3 - Z^2 + (A^2P - B^2P^2 - BP)Z - A^2BP^2 = 0 \quad (17)$$

ここで、 A と B はそれぞれ次式で与えられている。

$$A^2 = a/R^2T^{2.5} \quad (18)$$

$$B = b/RT \quad (19)$$

以下に、三次方程式(17)を導き、この方程式の解を示していく。式(18)と式(19)より、 $A^2/B = a/bRT^{1.5}$ であり、 $b/V = (Pb/RT)/(PV/RT) = BP/Z$ である。これらと式(1)を組み合わせて変形していく。 $Z = PV/RT = (V/RT)\{[RT/(V-b)] - [a/T^{0.5}V(V+b)]\}$

$$= \frac{1}{1-(b/V)} - \frac{a}{bRT^{1.5}V\{1+(V/b)\}}$$

$$= \frac{1}{1-(BP/Z)} - \frac{A^2(BP/Z)}{B\{1+(BP/Z)\}}$$

$$= \frac{Z}{Z-BP} - \frac{A^2P}{Z+BP} \quad (20)$$

式(20)の両辺の分母を払って整理すると、最終的に式(17)が得られる。この3次方程式をカルダノの公式を用いて解く。

$Y = Z - (1/3)$, $p = -1$, $q = BP(A^2/B - BP - 1)$, $r = -(A^2/B)(BP)^2$ とおくと、 Y に関する次の3次方程式が得られる。

$$Y^3 + \{q - (1/3)\}Y + \{(q/3) + r - (2/27)\} = 0 \quad (21)$$

ここで、 $m = q - (1/3)$ 、 $n = (q/3) + r -$

$(2/27)$ と置いておき、式(21)の実数解の数について場合分けをしながら、その解を示す。

まず、 $\alpha Y^3 + \beta Y^2 + \gamma Y + \delta = 0$ ($\alpha \neq 0$)の解の判別式(D)は次のように与えられている。

$$D = -4\alpha\gamma^3 - 27\alpha^2\delta^2 - 4\beta^3\delta + \beta^2\gamma^2 + 18\alpha\beta\gamma\delta \quad (22)$$

(1) $D > 0$ の時、異なる3つの実数解が存在する。

(2) $D = 0$ の時、3つの実数解が存在するがその内の少なくとも2つは等しい。

(3) $D < 0$ の時、ただ1つの実数解が存在する。

すると、式(21)より $D = -4m^3 - 27n^2$ になるので $-4m^3 - 27n^2$ の符号で場合分けをすれば良いことになる。

(1) $-4m^3 - 27n^2 > 0$ の時

3つの解に i ($i = 0, 1, 2$)の添字を付けると、

$$Y_i = 2(-m/3)^{0.5} \cos\{(\theta/3) + 120i\} \quad (23)$$

角度の単位は度であり、 θ は次の逆余弦関数から計算する。

$$\theta = \cos^{-1}\{\pm(-27n^2/4m^3)^{0.5}\} \quad (24)$$

ここで、 \pm は $n > 0$ ならば負、 $n < 0$ ならば正にとる。なお、 $D > 0$ であるので m は0にはならない。

すると、圧縮係数は $Z_i = Y_i + (1/3)$ と求められる。実数解が3つ存在するが、この内の最大値が求めるべき圧縮係数である(澁江, 1995)。

(2) $-4m^3 - 27n^2 = 0$ の時

これは $m = 0$, $n = 0$ の場合にあたる。すると、 $Y = 0$ と求められる(従って、圧縮係数は $1/3$ である)。

(3) $-4m^3 - 27n^2 < 0$ の時

$$Y = [(-n/2) + \{(n^2/4) + (m^3/27)\}^{0.5}]^{1/3} + [(-n/2) - \{(n^2/4) + (m^3/27)\}^{0.5}]^{1/3} \quad (25)$$

従って、 $Z = Y + (1/3)$ である。

3. プログラムの概要

プログラムはN₈₈日本語BASICで書かれたもので、そのリストをList AとList Bに示す(プログラム名はBOWERS and HELGESON, 1985の命名をそのまま踏襲している)。List Aは流体の密度を求めるプログラム("DENFIND")であり、List Bは流体中の各成分の逃散能と逃散能係数を求めるプログラム("FUGCO")である。また、各プログラムを用いた場合の入力例と出力例を"DENFIND"についてはList Cに、"FUGCO"についてはList Dに示す。

3.1 DENFIND

本プログラムはある圧力で組成と密度(g/cm^3)の入力値に対応する温度を求めるものである。このプログラムでは次の値を入力する。(1) 密度と組成値を計算する

組の数をline 1300 で入力する。(2) 密度の上限(maximum density) と下限(minimum density)、及びその変化幅(density increment)をline 1400 からline 1500 で入力する。(3) 組成 (X_{CO_2} と X_{NaCl}) をline 1600 からline 1650 で入力する。すると、 X_{H_2O} は $X_{H_2O} = 1 - X_{CO_2} - X_{NaCl}$ より計算できよう。これらの組成はY(1), Y(2), Y(3)で表されており、Y(1)がCO₂、Y(2)がH₂O、Y(3)がNaClのモル分率にそれぞれ対応している。

本プログラムでは圧力を500 barsから2000 bars まで500 bars刻みで変化させているが、line 2200, (PBの値は気圧を単位とする圧力を示す) とline 2100, (500 bar きざみ計算させている) を変えれば任意の圧力条件で計算可能になる。もし、300°C 以上600°C 以下の範囲で温度を見つけないことができなければ (line 2300 のTAとTBの値が温度範囲の最低値と最高値の絶対温度である)、解を求めることができなかつたと印刷される (line 5000)。

サブルーティンのREDKW はa とb の値から式(21)を解いてZ を計算するためのものである。そして式(16)を用いて(line 7450) 流体1 モルあたりの体積を計算している。また、line 5450から7550 にかけて示されているサブルーティンのTRKMIXでは式(2) と式(3) で与えられているパラメーター (a とb) を計算した後、逃散能係数を式(15)から求めている。このサブルーティンでは、H₂O あるいはNaClのモル分率が小さい場合、いくつかの近似が行われているので注意を要する。(1) まず、 $100-w$ が 10^{-6} より小さい (つまり、 X_{H_2O} が X_{NaCl} に比べて極めて小さい) 時、

$$(\partial n.b / \partial n_{NaCl}) = b_{H_2O}$$

$$(\partial n^2 a / \partial n_{NaCl}) = 2X_{NaCl} a_{H_2O}$$

と置いている。(2) w が 10^{-8} より小さい時、

$$(\partial n.b / \partial n_{H_2O}) = b_{H_2O}$$

と置いている。さらに、 X_{H_2O} が 10^{-8} より小さい場合、

$$(\partial n.b / \partial n_{NaCl}) = b_{H_2O}$$

$$(\partial n^2 a / \partial n_{NaCl}) = 2X_{CO_2} \{ (a_{H_2O} a_{CO_2})^{0.5} + 0.5R^2 T^{2.5} K \}$$

と置いている。NaClに関する偏微分式が原報で示されている式とは一致しないので注意する必要がある。

実際の計算では、組成から計算できる密度の値 (line 2050 で計算) をVOとして、温度をTAからTBまで変化させて密度を計算し (計算値をVLとする)、収束条件 (line 2950 からline 3050 あるいはline 3550 からline 3650) を満たす温度を探している。この収束判定は温度T1でのVLの計算値(VL₁) とT1より高温の温度T2 (あるいはT3) でのVLの計算値 (VL₂ あるいはVL₃) の間に、VL₁ < VO < VL₂ あるいはVL₁ < VO < VL₃ の関係が成立して、その上に次の不等式が成立する場合にあたる。

$$-0.001 + |0.5(VL_1 + VL_2) - VO| < VL_2 - VL_1$$

$$< 0.001 + |0.5(VL_1 + VL_2) - VO|$$

あるいは

$$-0.001 + |0.5(VL_1 + VL_3) - VO| < VL_3 - VL_1$$

$$< 0.001 + |0.5(VL_1 + VL_3) - VO|$$

この収束判定の際に、T1とT2 (あるいはT3) の温度差を小さくさせて T1を計算結果としている。

3. 2 FUGCO

本プログラムでは温度と圧力を指定して、様々な組成の流体について各成分の逃散能と逃散能係数を計算する。入力する値は次の通りである。(1) 温度の上限(T2)と下限(T1)及び温度の変化幅(TI)をline 10500からline 10750で入力する。(2) 圧力の上限(P2)と下限(P1)、及び圧力の変化幅(PI)を次にline 10650からline 10750で入力する。すると、逃散能係数をGAM、逃散能をFとして計算結果が印刷される。組成については、 X_{NaCl} の値を0 から0.06まで0.02きざみで変化させるとともに、 $X_{H_2O} / (X_{H_2O} + X_{CO_2})$ を0 から1 まで0.1 きざみで変化させている。ただし、 X_{CO_2} が大きくなって、 $X_{NaCl} / (X_{H_2O} + X_{NaCl}) > 0.14235$, (NaClの重量% にして35 wt%以上) になると解を求めることができなかつたと印刷される。もし、組成の変化幅を変える場合には、line 11800からline 12050を変えれば良い。

4. まとめ

本報告では、H₂O-CO₂-NaCl系流体に関する密度や各成分の逃散能を求めるプログラムを記載した。このプログラムはN₈₈-BASIC で書かれており、ユーザーが研究室でパソコンを用いて容易に利用できるものである。筆者はPC-9801RX 上で計算しているが、LIST CやLIST Dで示した計算はいずれも数分で可能である。

5. 引用文献

- BOWERS, T. S. and HELGESON, H. C. (1983) Calculation of the thermodynamic and geochemical consequences of nonideal mixing in the system H₂O-CO₂-NaCl on phase relations in geologic systems: equation of state for H₂O-CO₂-NaCl fluids at high pressures and temperatures. *Geochim. Cosmochim. Acta*, **47**, 1247-1275.
- BOWERS, T. S. and HELGESON, H. C. (1985) FORTRAN programs for generating fluid inclusion isochores and fugacity coefficients for the system H₂O-CO₂-NaCl at high pressures and temperatures. *Comput. & Geosci.*, **11**, 203-213.
- DE SANTIS, R., BREEDVELD, G. J. F. and

- PRAUSNITZ, J. M. (1974) Thermodynamic properties of aqueous gas mixtures at advanced pressures. *Ind. Eng. Chem. Process Des. Develop.*, **13**, 374-377.
- FERRY, J. M. and BAUMGARTNER, L. (1987) Thermodynamic models of molecular fluids at the elevated pressures and temperatures of crustal metamorphism. In CARMICHAEL, I. S. E. and EUGSTER, H. P. eds., *Reviews in Mineralogy*, **17**, Mineralogical Society of America, Washington, 323-365.
- GEHRIG, M. (1980) Phasegleichgewichte und PVT-Daten ternärer Mischungen aus Wasser, Kohlendioxide und Natriumchlorid bis 3 kbar und 550° C. PhD thesis, Univ. Karlsruhe, 109p.
- HOLLOWAY, J. R. (1977) Fugacity and activity of molecular species in supercritical fluids. In FRASER, D. G. ed., *Thermodynamics in Geology*, D. Reidel Pub. Co., Boston, 161-181.
- REDLICH, O. and KWONG, J. N. S. (1949) An equation of state. Fugacities of gaseous solution. *Chem. Rev.*, **44**, 233-244.
- ROEDDER, E. (1984) Fluid Inclusions. *Reviews in Mineralogy*, **14**, Mineralogical Society of America, Washington, 644p.
- 澁江 靖弘 (1990) H₂O-CO₂-NaCl系における fugacity coefficientの計算のための BASIC プログラム. *鉱山地質*, **40**, 45-46.
- 澁江 靖弘 (1993) いくつかのタングステン鉱床に関する流体包有物の研究. *兵庫教育大学研究紀要*, **13**, 第三分冊, 43-55.
- 澁江 靖弘 (1995) 流体の状態方程式: Redlich-Kwong Equation of State. *資源地質* (印刷中).

List A. Program DENFIND

```

1000 REM DENFIND
1040 LPRINT CHR$(27);"N"; WIDTH LPRINT 78
1050 LPRINT"X: mole fraction"
1060 LPRINT"Suffix 1: CO2, 2: H2O, 3: NaCl"
1100 DEFDBL A,B,C,D,F,G,H,P,Q,R,M,N,T,W,X,Y,Z
1150 DIM H(5,5),C(5,5),A(5,5),B(5,5),CH(10),PA(20)
1200 DIM PB(20),DS(200),DL(200),DD(200),QR(200,4)
1300 INPUT "Input number of data sets";DN
1350 FOR I=1 TO DN
1400 INPUT"Input minimum density";DS(I)
1450 INPUT"Input maximum density";DL(I)
1460 DL(I)=DL(I)+.000001#
1500 INPUT"Input density increment";DD(I)
1550 PRINT"Input mole fractions of CO2 and NaCl"
1600 INPUT"XCO2"; QR(I,1)
1650 INPUT"XNaCl";QR(I,3)
1700 QR(I,2)=1#-QR(I,1)-QR(I,3)
1750 NEXT I
1800 VV=0
1850 VV=VV+1
1900 D1=DS(VV):D2=DL(VV):DI=DD(VV):X(1)=QR(VV,1)
1910 X(2)=QR(VV,2):X(3)=QR(VV,3)
1950 LPRINT USING"XCO2=#.#### ";X(1);
1960 LPRINT USING"XH2O=#.#### ";X(2);
1970 LPRINT USING"XNaCl=#.####";X(3)
2000 D=D1
2050 VWO=X(1)*44.011#+X(2)*18.016#+X(3)*58.448#
2060 VO=VWO/D
2100 FOR LL=1 TO 4
2150 II=LL
2200 PB(LL)=500*LL
2250 PA(LL)=PB(LL)/1.01325#
2300 TA=573.15# :TB=873.15#
2350 TI=16
2400 XR=0
2450 TA=TA+.000000000001#:TB=TB+.000000000001#
2500 T1=TA
2550 T2=T1+TI
2600 T=T1
2650 GOSUB *TRKMIX
2700 F1=VL-VO
2750 IF ABS(F1)<.001# THEN GOTO 4300
2800 T=T2
2850 GOSUB *TRKMIX
2900 F2=VL-VO
2910 FF12=ABS(F1-F2)-ABS(F1+F2)
2950 IF ABS(FF12)<.001# THEN GOTO 4600
3000 IF FF12<0# THEN GOTO 4750
3050 IF FF12=0# THEN GOTO 4600
3100 IN=0
3150 T4=T1
3200 T3=(T4+T2)/2#
3250 T1=T3-T4
3300 T=T3
3350 GOSUB *TRKMIX
3400 F3=VL-VO
3450 IN=IN+1
3500 IF IN>20 GOTO 4450
3510 FF13=ABS(F1-F3)-ABS(F1+F3)
3550 IF ABS(FF13)<.001# THEN GOTO 4450
3600 IF FF13<0# THEN GOTO 4000
3650 IF FF13=0# THEN GOTO 4450
3700 T4=T3-TI
3750 T2=T2-TI
3800 T=T4
3850 GOSUB *TRKMIX
3900 F1=VL-VO
3950 GOTO 3200
4000 T4=T3
4050 T2=T2
4100 T=T4
4150 GOSUB *TRKMIX
4200 F1=VL-VO
4250 GOTO 3200
4300 XR=T1-273.15#
4350 LPRINT USING"D(g/cm^3)=#.#### ";D;
4360 LPRINT USING"P(bar)=####.## ";PB(II);
4365 LPRINT USING"T(°C)=###.###";XR
4400 GOTO 5050
4450 XR=T3-273.15#
4500 LPRINT USING"D(g/cm^3)=#.#### ";D;
4505 LPRINT USING"P(bar)=####.## ";PB(II);
4510 LPRINT USING"T(°C)=###.###";XR
4550 GOTO 5050
4600 XR=T1-273.15#
4650 LPRINT USING"D(g/cm^3)=#.#### ";D;
4655 LPRINT USING"P(bar)=####.## ";PB(II);
4660 LPRINT USING"T(°C)=###.###";XR
4700 GOTO 5050
4750 T1=T2
4800 T2=T2+TI
4850 IF T2<TB GOTO 2600
4900 IF T1>=TB THEN GOTO 5000
4950 T2=TB : GOTO 2600
5000 LPRINT " Root was not";
5010 LPRINT USING" found at ####.## (bars)";PB(II)
5050 NEXT LL
5100 D=D+DI
5150 IF D=<D2 THEN GOTO 2050
5200 IF VV=<DN-1 THEN GOTO 1850
5250 END
5300 REM DMIN=D1: DMAX=DD2: DINC=DI
5310 REM XCO2=X(1): XNaCl=X(3)
5350 REM DENS=D: VOL=VL: VOD=VO
5360 REM PBAR=PB: PAT=PA: XMIN=T1: XMAX=T2
5400 REM XINT=TI: XRT=XR: FCTN1=F1
5410 REM FCTN2=F2: FCTN3=F3

```

```

5450 *TRKMIX
5500 FOR I=1 TO 3:Y(I)=X(I):NEXT I
5550 P=PA(I)
5600 R=82.05#
5650 TC=T-273.15#
5700 R2=R*R*T*SQR(T)
5750 R1=R*T*SQR(T)
5800 IF Y(3)<1E-08 THEN GOTO 6000
5850 REM NaCl
5900 W=100#*Y(3)*58.488#
5910 W=W/(Y(3)*58.488#+Y(2)*18.016#)
5950 GOTO 6100
6000 W=0#
6050 REM A, B parameters for H2O
6100 BH=14.6#
6150 BN=-.04420283#
6200 AH=4.881243#+.1823047#*.01*TC
6205 AH=AH-.1712269#*.0001*TC*TC
6210 AH=AH+.6479419#*.00000001#*TC*TC*TC
6250 A1=.02636494#-.536994#*.001*TC
6255 A1=A1+.2687074#*.00001*TC*TC
6260 A1=A1-.4321741#*.00000001#*TC*TC*TC
6300 A2=.6802827#*.01-.948023#*.0001*TC
6310 A2=A2+.3770339#*.000001*TC*TC
6315 A2=A2-.5075318#*.000000001#*TC*TC*TC
6350 A3=.5235827#*.0001-.3505272#*.0000001*TC
6400 H(2,1)=BH+W*BN
6450 AN=(-8.05658#)*EXP((-982646#)*.01#*TC)
6500 HH12=50.70033#*EXP((-982646#)*.01#*TC)
6510 H(1,2)=111.3057#+HH12
6550 H(1,1)=EXP(AH+W*A1+W*W*A2+W*W*W*A3)
6555 H(1,1)=H(1,1)*1000000#
6600 H(1,3)=H(1,2)*1000000#
6650 AN=AN*1000000#
6700 H(1,4)=H(1,3)+W*AN
6750 REM CO2 parameter
6800 C(1,1)=46000000#
6850 C(2,1)=29.7#
6900 C(1,2)=73.03#-.0714#*TC
6910 C(1,2)=C(1,2)+2.157#*.00001#*TC*TC
6950 C(1,2)=C(1,2)*1000000#
7000 XXK=(-11.071#)+5953#/T-2.746#*1000000#/T/T
7005 XXK=XXK+4.646#*100000000#/T/T/T
7010 XK=EXP(XXK)
7050 HC(1)=XK*.5#*R2
7100 HC(1)=HC(1)+SQR(C(1,1)*H(1,1))
7150 Y(5)=Y(2)+Y(3)
7200 B(1,1)=C(2,1)*Y(1)+H(2,1)*Y(5)
7250 A(1,1)=Y(1)*Y(1)*C(1,2)+Y(5)*Y(5)*H(1,4)
7260 A(1,1)=A(1,1)+2#*Y(1)*Y(5)*HC(1)
7300 A(1,2)=A(1,1)/B(1,1)/R1
7350 B(1,2)=B(1,1)/R/T
7400 GOSUB *REDKW
7450 VL=R*T*Z/P
7500 FOR I=1 TO 5 : FOR J=1 TO 5
7510 H(J,I)=0#:C(J,I)=0#:B(J,I)=0#
7515 A(J,I)=0#:HC(I)=0#:Y(I)=0#
7520 NEXT J : NEXT I
7550 RETURN
7600 REM AZCO2=C(1,1):BCO2=C(2,1):TCEL=TC
7610 REM R2T=R2:RT=R1:AZH2O=H(1,1):AH2O=H(1,2)
7650 REM ANaCl=AN:AH2O=H(1,3):ACO2M=C(1,2)
7660 REM CO2H2O=CH(1,1):BSUM=B(1):ASUM=A(1,1)
7665 REM BMIX=B(1,2):A2BMIX=A(1,2):HC=CO2H2O
7700 REM CHI1=CH(1):CHI2=CH(2):CHI3=CH(3)
7750 *REDKW
7800 B=B(1,2):AB=A(1,2):BP=B*P:TH=1#/3#
7950 IF AB<0# THEN AB=.001#
8000 RR=(-1#)*AB*BP*BP
8050 QQ=BP*(AB-BP-1#)
8100 XN=(QQ/3#)+RR-2#/27#
8150 XM=QQ-1#/3#
8200 AR=(XN*XN/4#)+(XM*XM*XM/27#)
8250 IF AR>0# THEN GOTO 9350
8300 IF AR<0# THEN GOTO 8500
8350 FP=1#:Z=1#
8450 RETURN
8500 PH=SQR(XN*XN*27#/4#)/(-1#)/XM/XM/XM)
8550 PH=SQR(1#-COS(PH)*COS(PH))/COS(PH)
8600 IF XN>0# THEN PH=PH*(-1#)
8650 PH=(ATN(PH))/3#
8700 FC=2#*SQR(XM/3#)/(-1#)
8750 R1=FC*ABS(COS(PH))
8800 R2=FC*ABS(COS(PH+2.094395103#))
8850 R3=FC*ABS(COS(PH+4.188790206#))
8900 PH=3#*PH:RH=R2
9000 IF R1>R2 THEN RH=R1
9050 IF R3>RH THEN RH=R3
9100 IF RH=R1 THEN TA=0#
9150 IF RH=R2 THEN TA=2.094395103#
9200 IF RH=R3 THEN TA=4.188790206#
9250 Z=RH+1#/3#
9300 RETURN
9350 XX=SQR(AR)
9400 F=1#
9450 N2=XN/2#/-1#)
9500 MM=N2+XX
9550 IF MM<0# THEN F=-1#
9600 MM=LOG(ABS(MM))/3#
9650 MM=EXP(MM)*F
9700 F=1#
9750 NN=N2-XX
9800 IF NN<0# THEN F=-1#
9850 NN=LOG(ABS(NN))/3#
9900 NN=EXP(NN)*F
9950 Z=MM+NN+TH
10000 RETURN

```

List B. Program FUGCO

```

10000 REM FUGCO
10100 LPRINT CHR$(27);"Q";: WIDTH LPRINT 130
10200 LPRINT"X: mole fraction"
10205 LPRINT"F: fugacity"
10206 LPRINT"GAM: fugacity coefficient"
10210 LPRINT"Suffix 1: CO2, 2: H2O, 3: NaCl"
10250 DEFDBL A,B,C,D,F,G,H,P,Q,R,M,N,T,V,W,X,Y,Z
10300 DIM H(5,5), C(5,5), A(5,5), B(5,5), CH(10)
10400 SP$(1)="CO2" : SP$(2)="H2O" : SP$(3)="NaCl"
10450 PRINT "X denotes mole fraction"
10460 PRINT "GM denotes fugacity coefficient"
10470 PRINT "F denotes fugacity"
10500 INPUT"Input minimum temperature(°C)"; T1
10550 INPUT"Input maximum temperature(°C)"; T2
10560 T2=T2+.1#
10600 INPUT"Input temperature increment"; TI
10650 INPUT"Input minimum pressure(bars)"; P1
10700 INPUT"Input maximum pressure(bars)"; P2
10750 INPUT"Input pressure increment"; PI
10800 TC=T1
10850 T=TC+273.15#
10900 PB=P1
10950 PA=PB/1.01325#
11000 LPRINT
11050 LPRINT USING"###.###(°C) ";TC;
11060 LPRINT USING"#.#####(kb)";PB/1000#
11100 LPRINT
11150 Y(3)=0#
11200 FOR I=1 TO 11
11250 Y(1)=.1#*(I-1#)
11300 Y(2)=1#-Y(1)
11350 GOSUB *TRKMIX
11400 FOR JJ=1 TO 3
11450 F(JJ)=Y(JJ)*GM(JJ)*PB
11500 NEXT JJ
11550 FOR KJ=1 TO 3
11600 LPRINT USING "& & ";SP$(KJ);
11605 LPRINT USING "X=#.#### ";Y(KJ);
11610 LPRINT USING "GM=#.#####^";GM(KJ);
11620 LPRINT USING "F=#.#####^";F(KJ)
11650 NEXT KJ
11700 LPRINT
11750 NEXT I
11800 Y(3)=0#
11850 Y(3)=Y(3)+.02#
11900 IF Y(3)>.07# THEN GOTO 13100
11950 FOR I=1 TO 11
12000 Y(1)=(I-1#)*.1#*(1#-Y(3))
12050 Y(2)=1#-Y(1)-Y(3)
12100 GOSUB *TRKMIX
12150 FL=0#
12200 RA=Y(3)/(Y(3)+Y(2))
12250 IF RA>.14235# THEN FL=1#
12300 FOR JN=1 TO 3
12350 IF FL=1# THEN GM(JN)=0#
12400 F(JN)=Y(JN)*GM(JN)*PB
12450 NEXT JN
12500 IF GM(1)=0# GOTO 12750
12550 FOR JM=1 TO 3
12600 LPRINT USING"& & ";SP$(JM);
12605 LPRINT USING"X=#.#### ";Y(JM);
12610 LPRINT USING"GM=#.#####^";GM(JM);
12620 LPRINT USING"F=#.#####^";F(JM)
12650 NEXT JM
12700 GOTO 12950
12750 SS$="Root was not found"
12800 FOR JM=1 TO 3
12850 LPRINT USING"& & ";SP$(JM);
12855 LPRINT USING"X=#.#### ";Y(JM);
12860 LPRINT USING"& & ";SS$
12900 NEXT JM
12950 LPRINT
13000 NEXT I
13050 GOTO 11850
13100 PB=PB+PI
13150 IF PB=<P2 GOTO 13250
13200 GOTO 13350
13250 PA=PB/1.01325#
13300 GOTO 11000
13350 TC=TC+TI : PB=P1
13400 IF TC=<T2 THEN GOTO 10850
13450 END
13500 REM TMIN=T1:TMAX=T2:TINC=TI
13510 REM PMIN=P1:PMAX=P2:PINC=PI:PBAR=PB
13550 REM PAT=PA: GAM=GM: IFLAG=FL: RAT=RA
13600 *TRKMIX
13650 P=PA
13700 R=82.05#
13750 TC=T-273.15#
13800 R2=R*R*T*SQR(T)
13850 R1=R*T*SQR(T)
13900 IF Y(3)<.0000001# THEN GOTO 14100
13950 REM NaCl
14000 WWW=Y(3)*58.448#+Y(2)*18.016#
14010 W=100#*Y(3)*58.448#/WWW
14050 GOTO 14350
14100 W=0#
14150 REM A, B parameters for H2O
14200 BH=14.6#
14250 BN=-.04420283#
14300 AH=4.881243#+.1823047#*.01#*TC
14310 AH=AH-.1712269#*.0001#*TC*TC
14320 AH=AH+.6479419#*.0000001#*TC*TC*TC
14350 A1=.02636494#-.536994#*.001#*TC
14360 A1=A1+.2687074#*.00001#*TC*TC
14370 A1=A1-.4321741#*.0000001#*TC*TC*TC

```

```

14400 A2=.6802827*.01*- .948023*.0001*TC
14410 A2=A2+.3770339*.000001*TC*TC
14420 A2=A2-.5075318*.00000001*TC*TC*TC
14450 A3=.5235827*.0001*- .3505272*.0000001*TC
14500 REM H(2,1)=BH20
14550 H(2,1)=BH+W*BN
14600 REM ANaCl=AN
14650 AN=(-8.05658)*EXP((- .982646)*.01*TC)
14700 REM AH20=H(1,3)
14750 HH12=50.70033*EXP((- .982646*.01*TC))
14760 H(1,2)=111.3057*HH12
14800 REM AZH20=H(1,1)
14850 HH11=EXP(AH+W*A1+W*W*A2+W*W*W*A3)
14860 H(1,1)=HH11*1000000#
14900 H(1,3)=H(1,2)*1000000#
14950 AN=AN*1000000#
15000 H(1,4)=H(1,3)+W*AN
15050 REM CO2 parameter
15100 C(1,1)=46000000#
15150 C(2,1)=29.7#
15200 C(1,2)=73.03*- .0714*TC+2.157*.00001*TC*TC
15250 C(1,2)=C(1,2)*1000000#
15300 XXK=(-11.071#)+(5953#/T)
15310 XXK=XXK-(2.746*1000000#/T/T)
15315 XXK=XXK+(4.646*10000000#/T/T/T)
15320 XK=EXP(XXK)
15350 HC(1)=XK*.5*#*R2
15400 HC(1)=HC(1)+SQR(C(1,1)*H(1,1))
15450 Y(5)=Y(2)+Y(3)
15500 B(1,1)=C(2,1)*Y(1)+H(2,1)*Y(5)
15550 A(1,1)=Y(1)*Y(1)*C(1,2)+Y(5)*Y(5)*H(1,4)
15555 A(1,1)=A(1,1)+2*Y(1)*Y(5)*HC(1)
15600 A(1,2)=A(1,1)/B(1,1)/R1
15650 B(1,2)=B(1,1)/R/T
15700 GOSUB *REDKW
15750 VL=R*T*Z/P
15800 REM Calculation of fugacity coefficient
15850 CH(1)=LOG(VL/(VL-B(1,1)))
15900 CH(2)=LOG((VL+B(1,1))/VL)
15950 CH(3)=B(1,1)/(VL+B(1,1))
16000 DW=58.448#/18.016#/100#
16050 CW=18.016#/58.448#/100#
16100 IF ABS(W-100#)<.000001# THEN GOTO 16200
16150 GOTO 16500
16200 BI(1)=C(2,1)
16250 BI(2)=H(2,1)-CW*BN*W*W*Y(5)/Y(3)
16300 BI(3)=H(2,1)
16350 AI(1)=2*#*Y(1)*C(1,2)+2*#*Y(5)*HC(1)
16360 AI(2)=2*#*Y(5)*H(1,4)+2*#*Y(1)*HC(1)
16367 AI(2)=AI(2)-W*W*CW*Y(5)*AN*Y(5)/Y(3)
16368 AAI2=Y(1)*Y(5)*SQR(C(1,1)*H(1,1))*CW
16369 AAI2=AAI2*(A1+2*#*W*A2+3*#*W*W*A3)*W*W/Y(3)
16370 AI(2)=AI(2)-AAI2
16400 AI(3)=2*#*Y(3)*H(1,4)
16450 GOTO 17450
16500 BI(1)=C(2,1)
16550 IF W<.00000001# THEN GOTO 16950
16600 BI(2)=H(2,1)-CW*BN*W*W*Y(5)/Y(3)
16650 BBI3=DW*BN*(100#-W)*(100#-W)*Y(5)/Y(2)
16660 BI(3)=H(2,1)+BBI3
16700 AI(1)=2*#*Y(1)*C(1,2)+2*#*Y(5)*HC(1)
16750 AI(2)=2*#*Y(5)*H(1,4)+2*#*Y(1)*HC(1)
16752 AI(2)=AI(2)-CW*W*W*Y(5)*AN*Y(5)/Y(3)
16753 AAI2=Y(1)*Y(5)*SQR(C(1,1)*H(1,1))*CW
16754 AAI2=AAI2*(A1+2*#*W*A2+3*#*W*W*A3)*W*W/Y(3)
16755 AI(2)=AI(2)-AAI2
16800 AI(3)=2*#*Y(5)*H(1,4)+2*#*Y(1)*HC(1)
16810 AAI3=(100#-W)*(100#-W)*AN*Y(5)*Y(5)/Y(2)
16815 AI(3)=AI(3)+DW*AAI3
16850 AAI3=(A1+2*#*W*A2+3*#*W*W*A3)*DW
16855 AAI3=AAI3*Y(1)*Y(5)*SQR(C(1,1)*H(1,1))
16860 AI(3)=AI(3)+AAI3*(100#-W)*(100#-W)/Y(2)
16900 GOTO 17450
16950 AI(1)=2*#*Y(1)*C(1,2)+2*#*Y(5)*HC(1)
17000 AI(2)=2*#*Y(5)*H(1,4)+2*#*Y(1)*HC(1)
17050 IF Y(2)<.00000001# THEN GOTO 17300
17100 AAI3=2*#*Y(5)*H(1,4)+2*#*Y(1)*HC(1)
17110 AAI3=DW*(100#-W)*AN*(100#-W)*Y(5)
17120 AI(3)=AAI3+AAI3*Y(5)/Y(2)
17150 AII3=Y(1)*Y(5)*SQR(C(1,1)*H(1,1))
17160 AII3=AAI3*(A1+2*#*W*A2+3*#*W*W*A3)*DW
17170 AI(3)=AI(3)+AII3*(100#-W)*(100#-W)/Y(2)
17200 BBI3=DW*BN*(100#-W)*(100#-W)*Y(5)/Y(2)
17210 BI(3)=H(2,1)+BBI3
17250 GOTO 17400
17300 AI(3)=2*#*Y(1)*HC(1)
17350 BI(3)=H(2,1)
17400 BI(2)=H(2,1)
17450 FOR J=1 TO 3
17500 CH(5)=BI(J)/(VL-B(1,1))
17550 CH(6)=(A(1,1)*BI(J))/(R1*B(1,1)*B(1,1))
17600 CH(7)=AI(J)/(R1*B(1,1))
17650 PH=CH(1)+CH(5)-CH(7)*CH(2)
17660 PH=PH+CH(6)*(CH(2)-CH(3))-LOG(Z)
17700 PH=EXP(PH)
17750 GM(J)=PH
17800 NEXT J
17850 C(1,2)=0# : HC(1)=0#
17900 RETURN
17950 REM AZCO2=C(1,1):BCO2=C(2,1):TCCL=TC
17960 REM R2T=R2:RT=R1:AZH20=H(1,1):AH20=H(1,2)
17970 REM BH20=H(2,1):AH20=H(2,1):AH20M=H(1,4)
18000 REM ANaCl=AN:ACO2M=C(1,2):CO2H2O=HC(1)
18010 REM BSUM=B(1,1):ASUM=A(1,1)
18015 REM BMIX=B(1,2):A2BMIX=A(1,2)
18050 REM CHI1=CH(1):CHI2=CH(2):CHI3=CH(3)
18100 *REDKW
18150 B=B(1,2):AB=A(1,2)

```



```

18200 BP=B*P
18250 TH=1#/3#
18300 IF AB<0# THEN AB=.001#
18350 RR=(-1#)*AB*BP*BP
18400 QQ=BP*(AB-BP-1#)
18450 XN=(QQ/3#)+RR-2#/27#
18500 XM=QQ-1#/3#
18550 AR=(XN*XN/4#)+(XM*XM*XM/27#)
18600 IF AR>0# THEN GOTO 19700
18650 IF AR<0# THEN GOTO 18850
18700 FP=1#
18750 Z=1#
18800 RETURN
18850 PH=SQR(XN*XN*27#/4#*(-1#)/XM/XM/XM)
18900 PH=SQR(1#-COS(PH)*COS(PH))/COS(PH)
18950 IF XN>0# THEN PH=PH*(-1#)
19000 PH=(ATN(PH))/3#
19050 FC=2#*SQR(XM/3#*(-1#))
19100 R1=FC*ABS(COS(PH))
19150 R2=FC*ABS(COS(PH+2.094395103#))
19200 R3=FC*ABS(COS(PH+4.188790206#))
19250 PH=3#*PH
19300 RH=R2
19350 IF R1>RH THEN RH=R1
19400 IF R3>RH THEN RH=R3
19450 IF RH=R1 THEN TA=0#
19500 IF RH=R2 THEN TA=2.094395103#
19550 IF RH=R3 THEN TA=4.188790206#
19600 Z=RH+1#/3#
19650 RETURN
19700 X=SQR(AR)
19750 F=1#
19800 N2=XN/2#*(-1#)
19850 MM=N2+X
19900 IF MM<0# THEN F=-1#
19950 MM=LOG(ABS(MM))/3#
20000 MM=EXP(MM)*F
20050 F=1#
20100 NN=N2-X
20150 IF NN<0# THEN F=-1#
20200 NN=LOG(ABS(NN))/3#
20250 NN=EXP(NN)*F
20300 Z=MM+NN+TH
20350 RETURN
20400 REM A2B=AB:ARG=AR:PHI=PH:FAC=FC:THETA=TH
20410 REM XMM=MM:XMN=MN:YN2=N2:XNN=NN

```

List C. Input and output data on using DENFIND.

```

run
Input number of data sets? 2
Input minimum density? 0.7
Input maximum density? 0.8
Input density increment? 0.1
Input mole fractions of CO2 and NaCl
XCO2? 0.1
XNaCl? 0.2
Input minimum density? 0.6
Input maximum density? 0.7
Input density increment? 0.1
Input mole fractions of CO2 and NaCl
XCO2? 0.2
XNaCl? 0.05
OK

```

```

X: mole fraction
Suffix 1: CO2, 2: H2O, 3: NaCl
XCO2=0.1000  XH2O=0.8800  XNaCl=0.0200
D(g/cm^3)=0.7000  P(bar)= 500.00  T(°C)=348.969
D(g/cm^3)=0.7000  P(bar)=1000.00  T(°C)=408.203
D(g/cm^3)=0.7000  P(bar)=1500.00  T(°C)=470.188
D(g/cm^3)=0.7000  P(bar)=2000.00  T(°C)=535.813
D(g/cm^3)=0.8000  P(bar)= 500.00  T(°C)=306.313
D(g/cm^3)=0.8000  P(bar)=1000.00  T(°C)=350.094
D(g/cm^3)=0.8000  P(bar)=1500.00  T(°C)=394.969
D(g/cm^3)=0.8000  P(bar)=2000.00  T(°C)=441.219
XCO2=0.2000  XH2O=0.7700  XNaCl=0.0300
D(g/cm^3)=0.6000  P(bar)= 500.00  T(°C)=343.652
D(g/cm^3)=0.6000  P(bar)=1000.00  T(°C)=435.734
D(g/cm^3)=0.6000  P(bar)=1500.00  T(°C)=540.078
          Root was not found at 2000.00 (bars)
D(g/cm^3)=0.7000  P(bar)= 500.00  T(°C)=313.637
D(g/cm^3)=0.7000  P(bar)=1000.00  T(°C)=382.289
D(g/cm^3)=0.7000  P(bar)=1500.00  T(°C)=453.797
D(g/cm^3)=0.7000  P(bar)=2000.00  T(°C)=533.594

```

List D. Input on using FUGCO.

```

run
X denotes mole fraction
GM denotes fugacity coefficient
F denotes fugacity
Input minimum temperature(°C)? 500
Input maximum temperature(°C)? 500
Input temperature increment? 100
Input minimum pressure(bars)? 1000
Input maximum pressure(bars)? 1000
Input pressure increment? 1000
OK

```

List D-2. Output on using FUGCO.

X: mole fraction

F: fugacity

GM: fugacity coefficient

Suffix 1: CO₂, 2: H₂O, 3: NaCl

500.000(°C) 1.00000(kb)

CO ₂	X=0.0000	GM=+4.24663D+00	F=+0.0000D+00
H ₂ O	X=1.0000	GM=+4.38016D-01	F=+4.3802D+02
NaCl	X=0.0000	GM=+1.89675D-01	F=+0.0000D+00

CO ₂	X=0.1000	GM=+2.41628D+00	F=+2.4163D+02
H ₂ O	X=0.9000	GM=+4.49881D-01	F=+4.0489D+02
NaCl	X=0.0000	GM=+6.46436D-01	F=+0.0000D+00

CO ₂	X=0.2000	GM=+1.81660D+00	F=+3.6332D+02
H ₂ O	X=0.8000	GM=+4.72264D-01	F=+3.7781D+02
NaCl	X=0.0000	GM=+1.47152D+00	F=+0.0000D+00

CO ₂	X=0.3000	GM=+1.56004D+00	F=+4.6801D+02
H ₂ O	X=0.7000	GM=+4.96352D-01	F=+3.4745D+02
NaCl	X=0.0000	GM=+2.74520D+00	F=+0.0000D+00

CO ₂	X=0.4000	GM=+1.42866D+00	F=+5.7146D+02
H ₂ O	X=0.6000	GM=+5.20107D-01	F=+3.1206D+02
NaCl	X=0.0000	GM=+4.60266D+00	F=+0.0000D+00

CO ₂	X=0.5000	GM=+1.35397D+00	F=+6.7698D+02
H ₂ O	X=0.5000	GM=+5.43232D-01	F=+2.7162D+02
NaCl	X=0.0000	GM=+7.22616D+00	F=+0.0000D+00

CO ₂	X=0.6000	GM=+1.30927D+00	F=+7.8556D+02
H ₂ O	X=0.4000	GM=+5.65791D-01	F=+2.2632D+02
NaCl	X=0.0000	GM=+1.08451D+01	F=+0.0000D+00

CO ₂	X=0.7000	GM=+1.28231D+00	F=+8.9762D+02
H ₂ O	X=0.3000	GM=+5.87907D-01	F=+1.7637D+02
NaCl	X=0.0000	GM=+1.57424D+01	F=+0.0000D+00

CO ₂	X=0.8000	GM=+1.26673D+00	F=+1.0134D+03
H ₂ O	X=0.2000	GM=+6.09695D-01	F=+1.2194D+02
NaCl	X=0.0000	GM=+2.22635D+01	F=+0.0000D+00

CO ₂	X=0.9000	GM=+1.25891D+00	F=+1.1330D+03
H ₂ O	X=0.1000	GM=+6.31248D-01	F=+6.3125D+01
NaCl	X=0.0000	GM=+3.08271D+01	F=+0.0000D+00

CO ₂	X=1.0000	GM=+1.25664D+00	F=+1.2566D+03
H ₂ O	X=0.0000	GM=+6.52639D-01	F=+9.0572D-15
NaCl	X=0.0000	GM=+6.52639D-01	F=+0.0000D+00

CO ₂	X=0.0000	GM=+6.00610D+00	F=+0.0000D+00
H ₂ O	X=0.9800	GM=+4.37872D-01	F=+4.2911D+02
NaCl	X=0.0200	GM=+1.96065D-01	F=+3.9213D+00

CO ₂	X=0.0980	GM=+2.93886D+00	F=+2.8801D+02
H ₂ O	X=0.8820	GM=+4.40659D-01	F=+3.8866D+02
NaCl	X=0.0200	GM=+7.09930D-01	F=+1.4199D+01

CO ₂	X=0.1960	GM=+2.05490D+00	F=+4.0276D+02
H ₂ O	X=0.7840	GM=+4.59888D-01	F=+3.6055D+02
NaCl	X=0.0200	GM=+1.51491D+00	F=+3.0298D+01

CO ₂	X=0.2940	GM=+1.69949D+00	F=+4.9965D+02
H ₂ O	X=0.6860	GM=+4.83355D-01	F=+3.3158D+02
NaCl	X=0.0200	GM=+2.38516D+00	F=+4.7703D+01

CO ₂	X=0.3920	GM=+1.51989D+00	F=+5.9580D+02
H ₂ O	X=0.5880	GM=+5.09517D-01	F=+2.9960D+02
NaCl	X=0.0200	GM=+2.96084D+00	F=+5.9217D+01

CO ₂	X=0.4900	GM=+1.41564D+00	F=+6.9366D+02
H ₂ O	X=0.4900	GM=+5.40470D-01	F=+2.6483D+02
NaCl	X=0.0200	GM=+2.87040D+00	F=+5.7408D+01

CO ₂	X=0.5880	GM=+1.34995D+00	F=+7.9377D+02
H ₂ O	X=0.3920	GM=+5.80858D-01	F=+2.2770D+02
NaCl	X=0.0200	GM=+2.10764D+00	F=+4.2153D+01

CO ₂	X=0.6860	GM=+1.30678D+00	F=+8.9645D+02
H ₂ O	X=0.2940	GM=+6.37367D-01	F=+1.8739D+02
NaCl	X=0.0200	GM=+1.21244D+00	F=+2.4249D+01

CO ₂	X=0.7840	GM=+1.27874D+00	F=+1.0025D+03
H ₂ O	X=0.1960	GM=+7.09443D-01	F=+1.3905D+02
NaCl	X=0.0200	GM=+7.21162D-01	F=+1.4423D+01

CO ₂	X=0.8820	Root was not found	
H ₂ O	X=0.0980	Root was not found	
NaCl	X=0.0200	Root was not found	

CO ₂	X=0.9800	Root was not found	
H ₂ O	X=0.0000	Root was not found	
NaCl	X=0.0200	Root was not found	

CO2 X=0.0000 GM=+8.13018D+00 F=+0.0000D+00
 H2O X=0.9600 GM=+4.37471D-01 F=+4.1997D+02
 NaCl X=0.0400 GM=+2.01993D-01 F=+8.0797D+00

CO2 X=0.0960 GM=+3.43749D+00 F=+3.3000D+02
 H2O X=0.8640 GM=+4.39495D-01 F=+3.7972D+02
 NaCl X=0.0400 GM=+4.62850D-01 F=+1.8514D+01

CO2 X=0.1920 GM=+2.24630D+00 F=+4.3129D+02
 H2O X=0.7680 GM=+4.64035D-01 F=+3.5638D+02
 NaCl X=0.0400 GM=+6.69158D-01 F=+2.6766D+01

CO2 X=0.2880 GM=+1.79605D+00 F=+5.1726D+02
 H2O X=0.6720 GM=+4.95978D-01 F=+3.3330D+02
 NaCl X=0.0400 GM=+7.54150D-01 F=+3.0166D+01

CO2 X=0.3840 GM=+1.57358D+00 F=+6.0425D+02
 H2O X=0.5760 GM=+5.33159D-01 F=+3.0710D+02
 NaCl X=0.0400 GM=+7.33055D-01 F=+2.9322D+01

CO2 X=0.4800 GM=+1.44545D+00 F=+6.9381D+02
 H2O X=0.4800 GM=+5.75735D-01 F=+2.7635D+02
 NaCl X=0.0400 GM=+6.61216D-01 F=+2.6449D+01

CO2 X=0.5760 GM=+1.36555D+00 F=+7.8656D+02
 H2O X=0.3840 GM=+6.22525D-01 F=+2.3905D+02
 NaCl X=0.0400 GM=+5.99451D-01 F=+2.3978D+01

CO2 X=0.6720 GM=+1.31448D+00 F=+8.8333D+02
 H2O X=0.2880 GM=+6.69628D-01 F=+1.9285D+02
 NaCl X=0.0400 GM=+5.85859D-01 F=+2.3434D+01

CO2 X=0.7680 Root was not found
 H2O X=0.1920 Root was not found
 NaCl X=0.0400 Root was not found

CO2 X=0.8640 Root was not found
 H2O X=0.0960 Root was not found
 NaCl X=0.0400 Root was not found

CO2 X=0.9600 Root was not found
 H2O X=0.0000 Root was not found
 NaCl X=0.0400 Root was not found

CO2 X=0.0000 GM=+9.97719D+00 F=+0.0000D+00
 H2O X=0.9400 GM=+4.36852D-01 F=+4.1064D+02
 NaCl X=0.0600 GM=+2.07510D-01 F=+1.2451D+01

CO2 X=0.0940 GM=+3.79519D+00 F=+3.5675D+02
 H2O X=0.8460 GM=+4.41866D-01 F=+3.7382D+02
 NaCl X=0.0600 GM=+3.47190D-01 F=+2.0831D+01

CO2 X=0.1880 GM=+2.36379D+00 F=+4.4439D+02
 H2O X=0.7520 GM=+4.71050D-01 F=+3.5423D+02
 NaCl X=0.0600 GM=+4.32577D-01 F=+2.5955D+01

CO2 X=0.2820 GM=+1.84876D+00 F=+5.2135D+02
 H2O X=0.6580 GM=+5.06669D-01 F=+3.3339D+02
 NaCl X=0.0600 GM=+4.72155D-01 F=+2.8329D+01

CO2 X=0.3760 GM=+1.60092D+00 F=+6.0195D+02
 H2O X=0.5640 GM=+5.44899D-01 F=+3.0732D+02
 NaCl X=0.0600 GM=+4.91027D-01 F=+2.9462D+01

CO2 X=0.4700 GM=+1.46092D+00 F=+6.8663D+02
 H2O X=0.4700 GM=+5.84421D-01 F=+2.7468D+02
 NaCl X=0.0600 GM=+5.09182D-01 F=+3.0551D+01

CO2 X=0.5640 GM=+1.37509D+00 F=+7.7555D+02
 H2O X=0.3760 GM=+6.24121D-01 F=+2.3467D+02
 NaCl X=0.0600 GM=+5.37734D-01 F=+3.2264D+01

CO2 X=0.6580 Root was not found
 H2O X=0.2820 Root was not found
 NaCl X=0.0600 Root was not found

CO2 X=0.7520 Root was not found
 H2O X=0.1880 Root was not found
 NaCl X=0.0600 Root was not found

CO2 X=0.8460 Root was not found
 H2O X=0.0940 Root was not found
 NaCl X=0.0600 Root was not found

CO2 X=0.9400 Root was not found
 H2O X=0.0000 Root was not found
 NaCl X=0.0600 Root was not found

BASIC programs for the calculation of density and fugacity of H₂O-CO₂-NaCl fluid

Yasuhiro SHIBUE

The N₈-BASIC code is developed for the calculation of the density and the fugacity of the component gases in the system H₂O-CO₂-NaCl. This code is written on the basis of the FORTRAN code shown by BOWERS and HELGESON (1985). Using the present code for a personal computer, calculations of the fluid properties can be performed easily.